p-ISSN: 2184-044X e-ISSN: 2184-1489 ISBN: 978-989-35106-9-8 © 2024
https://doi.org/10.36315/2024v1end114

SIMULATION MODELING OF A PRODUCTION SYSTEM

Chavdar Kostadinov[1], & Ivanka Peeva[2]
[1] "G. S. Rakovski" National Defence College (Bulgaria)
[2] University of Ruse "Angel Kanchev" (Bulgaria)

Abstract

Modeling of production and communication systems is one of the modern trends in technical research, finding great practical application. Studying the features and application of simulation modeling by students in engineering specialties allows them to become familiar with real production tasks and the possibilities for their solution. This paper describes a research task considered in the training of students in the field of industrial production systems. A system made up of two production machines served by one robot is presented. Estimated performance values can be obtained and analyzed based on the results of simulation modeling of the system's performance under certain production conditions. For this purpose, the GPSS programming environment is used. Students are introduced to the programming language, basic GPSS operators, examine an example model or create a simulation model of the system under consideration. The obtained results in simulation modeling provide an opportunity for analysis and search for optimal solutions under set criteria for system efficiency. Solving such tasks would support the engineering training of students and their practical implementation.

Keywords: *Industrial robot, robotic technology module, simulation modeling, GPSS.*

1. Introduction

Simulation modeling allows solving many types of problems. For the purpose of teaching students of various master's courses in the field of industrial engineering, a production system consisting of two production machines that are served by one robot is considered. This is a commonly used robotic system in practice – a type of robotic technological module (RTM) that can work as a stand-alone unit or as part of a single system. This equipment can be considered as a mass service system, and the modeling done based on Queuing Theory (QT). The schematic diagram of such a production system is presented in figure 1.

Figure 1. Schematic of a production system.



The location of the machines in the assembly diagram of the robotic module can be different, determined by the design features and overall dimensions of the machines, as well as by the features of the occupied work space. In this example scheme, the two machines (M1 and M2) are symmetrically located around the robot (Robot), and it is possible to sequentially perform operations from a single technological process (conditionally divided between them) or both to implement the entire technological process in parallel. The difference between the two types of work organization is in the service cycle of the machines and the need for an intermediate device (MD) in the sequential operation of machines with different operating times.

2. Methodology

2.1. Determining the range of service time variation

In sequential operation, the workpiece arrives at the system input (ID). Then, if the robot is free, it takes it and places it in the working area of the first machine. After finishing the processing, the robot leaves it in the intermediate device. With the second machine free, the robot takes a part from the intermediate device and places it in the work area of the second machine. After finishing the processing, the robot places it in the output device (OD), which completes one cycle of operation of the production system with sequential action. The production cycle of a parallel-acting industrial system begins with the arrival of the part at the system input (ID). If the milling robot is free, it picks up and places the workpiece in the working area of the first machine M1, after which it starts processing it. The robot is free to service the second machine M2 by taking a part from the input device and placing it in the work area of the second machine. Each of the machines, having finished processing a part, gives a signal for service from the robot. The machines are serviced in the order in which requests are received or according to a set priority (Kostadinov, 2017).

For the purpose of the exercise, each of the students considers a technological process with certain operating times realized by the two machines. Using the GPSS simulation environment (Schreiber, 1980), a simulation model of the considered production system must be created, through which its main characteristics can be predicted, such as performance and load of the equipment used under certain operating conditions. This allows students to become familiar with solving practical tasks on the optimization of industrial production tasks. In order to determine and analyze the performance of the system under consideration, it is necessary to set appropriate machine service times. These times must correspond to the permissible relative intensity ρ for the system under consideration: $0.1 < \rho < 0.5$ (Kostadinov, 2017):

$$\rho=\lambda/\mu=t_R/t_M , \quad (1)$$

where $\lambda=1/t_M$, λ is intensity of service requests, t_M is the average working time of machines, $\mu=1/t_R$, μ is intensity of service, t_R is the average working time of the robot.

Dependency (1) is used to determine the service time change limits, as:

$$t_R=\rho*t_M \quad (2)$$

Given machine times and an allowable range for variation of ρ, the appropriate range for machine service time variation can be determined. For close machine times, it is appropriate to work with their average value.

$$t_{Rmin} = \rho_{min}\ t_M=0,1*t_M$$
$$t_{Rmax} = \rho_{max}\ t_M=0,5*t_M$$

Using values of service times in the specified range and set machine times, through simulation modeling based on QT, it is possible to obtain estimated values for the main characteristics of the considered system, to be used in the selection of a suitable robot and its technical parameters. Consideration of these questions has a certain practical significance for the education of students in the field of modern industrial systems.

2.2. Simulation modeling in a GPSS environment

Once the service time variation range is determined, one can proceed to program creation in the GPSS environment (Norenkov, 1994). It is necessary to acquaint the students with the features of the simulation software and the operators used to create a model of the considered real system. Some of these operators are (GPSS world reference manual, 2001):

GENERATE (Exponential (q, w, d)) – generation of transactions, time intervals between the occurrence of transactions distributed by the value of the function Exponential; the first transaction appears with a delay of q units of model time, all w transactions will be created, the priority of the transactions is equal to d.

SEIZE t – occupying the device t from the transactions arriving at its input; if a device is busy, the transaction is held in the queue for it;

RELEASE s – release of the device s from the serviced transaction;
QUEUE f – queue organization operator, the length of the queue f is increased by one;
DEPART j – queue organization operator, the length of queue j is reduced by one;
ADVANCE i, p – holds the transaction for a time determined by the contents of fields i and p;
TABULATE h7 – in the corresponding interval (7) of the histogram named h, a unit is added;
TERMINATE g - removes the transaction from the system, the content of its counter is decremented by g units, modeling ends if the content of the counter becomes equal to or less than zero.

An example model from the GPSS World software package [4] presented in figure 2.

Figure 2. Example program in the GPSS environment.

```
Listing
; GPSS World Sample File - ROBOTFMS.GPS
*******************************************************************
* Experimental Manufacturing Cell
* Two CNC machines and one Robot
* One arrival area and one finished parts area
*******************************************************************
RMULT 78863
Transit TABLE M1,100,100,20                ;Record lead time
*******************************************************************
        GENERATE (Exponential(1,0,150))    ;A job arrives
        QUEUE     One                      ;Arrival queue
        SEIZE     Robot                    ;Get the robot
        DEPART    One                      ;Depart the queue
        ADVANCE   8,1                      ;Robot grips the job
        ADVANCE   3                        ;Robot moves to machine 1
        ADVANCE   6,1                      ;Robot place the job
        RELEASE   Robot                    ;Free the robot
        QUEUE     Two                      ;Wait in next queue
        SEIZE     Machine1                 ;Get first machine
        DEPART    Two                      ;Depart the queue
        ADVANCE  (Exponential(1,0,100))    ;Process time
        RELEASE   Machine1                 ;Free machine 1
        QUEUE     Three                    ;Join queue for machine 2
        SEIZE     Robot                    ;Get the robot
        DEPART    Three                    ;Depart the queue
        ADVANCE   8,1                      ;Robot grips part
        ADVANCE   3                        ;Robot moves to machine 2
        ADVANCE   8,1                      ;Robot places the part
        RELEASE   Robot                    ;Free the robot
        QUEUE     Four                     ;Join queue machine 2
        SEIZE     Machine2                 ;Get machine 2
        DEPART    Four                     ;Depart the queue
        ADVANCE  (Exponential(1,0,80))     ;Process 2
        RELEASE   Machine2                 ;Free machine 2
        QUEUE     Five                     ;Queue for exit station
        SEIZE     Robot                    ;Get the robot
        DEPART    Five                     ;Depart the queue
        ADVANCE   8,1                      ;Robot grips the part
        ADVANCE   3                        ;Robot moves to exit
        ADVANCE   6,1                      ;Robot places the part
        RELEASE   Robot                    ;Free the robot
        TABULATE  Transit                  ;Transit time
        TERMINATE 1                        ;Job is completed
*******************************************************************
```

After running a certain number of simulations, the resulting data can be tracked in the GPSS generated report presented in figure 3. The data should be analyzed with a view to determining the credibility of the proposed simulation model and searching for opportunities to reach the set technical parameters of the system.

Figure 3. GPSS generated report.

```
              GPSS World Simulation Report - Untitled Model 1.2.1
               Friday, March 08, 2024 20:20:07
        START TIME           END TIME  BLOCKS  FACILITIES  STORAGES
             0.000           16744.929    34        3          0


           NAME                       VALUE
       FIVE                        10014.000
       FOUR                        10012.000
       MACHINE1                    10010.000
       MACHINE2                    10013.000
       ONE                         10007.000
       ROBOT                       10008.000
       THREE                       10011.000
       TRANSIT                     10006.000
       TWO                         10009.000

 LABEL              LOC  BLOCK TYPE     ENTRY COUNT CURRENT COUNT RETRY
                    1    GENERATE          109             0       0
                    2    QUEUE             109             0       0
                    3    SEIZE             109             0       0
                    4    DEPART            109             0       0
                    5    ADVANCE           109             0       0
                    6    ADVANCE           109             0       0
                    7    ADVANCE           109             0       0
                    8    RELEASE           109             0       0
                    9    QUEUE             109             6       0
                   10    SEIZE             103             0       0
                   11    DEPART            103             0       0
                   12    ADVANCE           103             1       0
                   13    RELEASE           102             0       0
                   14    QUEUE             102             0       0
                   15    SEIZE             102             0       0
                   16    DEPART            102             0       0
                   17    ADVANCE           102             0       0
                   18    ADVANCE           102             0       0
                   19    ADVANCE           102             0       0
                   20    RELEASE           102             0       0
                   21    QUEUE             102             0       0
                   22    SEIZE             102             0       0
                   23    DEPART            102             0       0
                   24    ADVANCE           102             1       0
                   25    RELEASE           101             0       0
                   26    QUEUE             101             0       0
                   27    SEIZE             101             1       0
                   28    DEPART            100             0       0
                   29    ADVANCE           100             0       0
                   30    ADVANCE           100             0       0
                   31    ADVANCE           100             0       0
                   32    RELEASE           100             0       0
                   33    TABULATE          100             0       0
                   34    TERMINATE         100             0       0

FACILITY ENTRIES UTIL.   AVE. TIME AVAIL. OWNER PEND INTER RETRY DELAY
 ROBOT     312    0.327      17.552  1      101    0    0     0      0
 MACHINE1  103    0.581      94.491  1      103    0    0     0      6
 MACHINE2  102    0.443      72.688  1      102    0    0     0      0

QUEUE MAX CONT. ENTRY ENTRY(0) AVE.CONT. AVE.TIME AVE.(-0) RETRY
 ONE              2    0    109     75     0.022      3.377     10.825   0
 TWO              9    6    109     41     0.849    130.432    209.075   0
 THREE            2    0    102     67     0.028      4.648     13.545   0
 FOUR             5    0    102     53     0.377     61.811    128.668   0
 FIVE             2    1    101     67     0.023      3.832     11.384   0

TABLE                     MEAN    STD.DEV.       RANGE    RETRY FREQUENCY CUM.%
 TRANSIT         384.804  207.036                          0
                                        _  -      100.000                2     2.00
                                 100.000  -      200.000               17    19.00
                                 200.000  -      300.000               20    39.00
                                 300.000  -      400.000               18    57.00
                                 400.000  -      500.000               19    76.00
                                 500.000  -      600.000               10    86.00
                                 600.000  -      700.000                6    92.00
```

```
     700.000  -      800.000             4    96.00
     800.000  -      900.000             0    96.00
     900.000  -     1000.000             3    99.00
    1000.000  -     1100.000             1   100.00
```

3. Results

For the purpose of practical work for simulation modeling of the robotic system presented in Figure 1, students may create their own program or use the one proposed in Figure 2. Machine operation and service times are set and a certain number of simulations are run and the results are summarized. For example, at specific time's 14 – 31 s, 10 simulations were made under the same general conditions (set volume of production 100 parts) and different time for the service robot to work, the results presented in the table were obtained. 1.

Maximum productivity is reached at service time t_R = 17.552 s, respectively Q_{PSh} = 9.355 units/hour. The general dependence is confirmed that as the cycle time increases, the productivity of the considered production system decreases, which confirms the credibility of the adopted model. Several simulations (1, 4, 7) are striking, where different results are obtained. Given certain ratios of work and service times, such results are possible. The presence of such cases in real production conditions requires an appropriate analysis of the factors leading to the relevant deviations and making correct decisions regarding the organization of work in the production system. During an exercise, it is appropriate to consider incoming proposals from the students regarding possible solutions to reach set values of the efficiency criteria of the system under consideration, to be verified by simulation. This would lead to the expansion of their practical experience, and the discussion of the performed activities - to the improvement of their teamwork skills.

Table 1. Results of conducted simulations.

№	t_R, s	T_{PS}, s	Q_{PS}, un/s	Q_{PSh}, un/h
1	14,622	610,715	0,0016	5,895
2	16,582	401,942	0,0025	8,957
3	17,552	384,804	0,0026	9,355
4	18,548	528,398	0,0019	6,813
5	19,576	525,166	0,0019	6,855
6	20,566	447,981	0,0022	8,036
7	22,596	645,927	0,0015	5,573
8	24,633	455,973	0,0022	7,895
9	27,677	572,252	0,0017	6,291
10	31,573	779,558	0,0013	4,618

4. Conclusion

The use of simulation modeling in the study of systems of machines or technical devices allows analyzing and predicting their various parameters under certain operating conditions. Knowledge of modern general-purpose simulation software (GPSS) and practical work with it allows expanding the professional competence of young specialists facilitates decision-making in real production tasks and leads to deepening of students' practical skills when performing engineering activities in practice.

References

GPSS world reference manual. (2001). Minuteman Software. Holly Springs, NC, USA. Retrieved from https://athena.ecs.csus.edu/~mitchell/csc148/gpssW/Reference%20Manual/rpreface.htm

Kostadinov, Ch. (2017). *Probabilistic modeling, analysis and performance forecasting of production systems with a parallel structure* (Doctoral dissertation, "Angel Kanchev" University of Ruse).

Schreiber, T. D. (1980). *GPSS modeling*. Moscow: Mechanical Engineering.

Norenkov, I. P. (1994). *CAD development*. Moscow: MGTU Baumann.